

A função procv

Observemos as tabelas abaixo:

The screenshot shows an Excel spreadsheet with two tables. The first table, 'Tabela 1 - Itens Vendidos', is located in rows 5 to 9. It has columns for 'Filme', 'Quantidade', 'Valor Unitário', and 'Valor Total'. The second table, 'Tabela 2 - Lista de Produtos', is located in rows 14 to 19. It has columns for 'Filme', 'Categoria', and 'Valor'. The 'Valor' column in Tabela 1 is empty, while the 'Valor' column in Tabela 2 contains numerical values.

Tabela 1 - Itens Vendidos			
Filme	Quantidade	Valor Unitário	Valor Total
No Limite	10		
Carros	12		
Naufrago	4		
Chamado	3		

Tabela 2 - Lista de Produtos			
Filme	Categoria	Valor	
Chamado	Suspense	R\$ 42,00	
Carros	Infantil	R\$ 20,00	
Galinha Pintadinha	Infantil	R\$ 20,00	
No Limite	Ação	R\$ 30,00	
Titanic	Romance	R\$ 18,00	

A primeira mostra a quantidade vendida de cada item de uma loja de DVDs. Nela precisaremos preencher as colunas com os valores unitários e o valor total de acordo com a quantidade. A segunda tabela também é uma lista de filmes com suas categorias discriminadas e o valor unitário de cada um.

Como sempre, damos exemplos com planilhas pequenas, mas vamos pensar em outras mais extensas e tentar resolver o problema de maneira mais automatizada. A ideia aqui é pegar as informações da tabela 2 e inserir na tabela 1, para depois calcularmos o valor total vendido de cada item.

Perceba que o item "Náufrago" não está discriminado na tabela 2. Se preenchermos manualmente, a sua célula de "Valor Unitário" ficaria em branco.

Utilizaremos um recurso do Excel que é a função `PROCV()` para preencher a tabela automaticamente. Implementando-a em uma tabela, ela buscará valores em outra auxiliar.

A função `PROCV()` funciona da seguinte forma:

`=PROCV(Valor_Procurado;Matriz_Tabela;Número_Coluna;Tipo_Procura)`

Sendo

- **Valor Procurado:** valor que será localizado na primeira coluna da tabela auxiliar;
- **Matriz_Tabela:** região de células que contém as informações que iremos buscar;
- **Número_Coluna:** Número da coluna da tabela auxiliar onde iremos buscar as informações;
- **Tipo_Procura:**
 - *0 ou False:* busca exata
 - *1 ou True:* busca por faixa numérica

Comecemos a implementar a função pensando no primeiro item:

- É procurado o item "No Limite" na tabela auxiliar (A6 na tabela 1): `=PROCV(A6; ; ;)`;
- É procurado na região da tabela 2 (região B15:D19): `=PROCV(A6;B15:D19; ;)`;
- As informações que queremos inserir estão na última coluna da tabela 2 (coluna 3): `=PROCV(A6;B15:D19;3;)`;
- Buscamos **exatamente** o nome do filme, ou seja, é uma busca exata (*0* ou *False*): `=PROCV(A6;B15:D19;3;0)` .

Aplicando a fórmula na primeira célula e arrastando com a Alça:

D8 : []

A B C D

1
2
3
4 **Tabela 1 - Itens Vendidos**
5 Filme Quantidade Valor Unitário Valor Total
6 No Limite 10 R\$ 30,00
7 Carros 12 R\$ 20,00
8 Naufrago 4 #N/D
9 Chamado 3 #N/D

10

11
12
13 **Tabela 2 - Lista de Produtos**
14 Filme Categoria Valor
15 Chamado Suspense R\$ 42,00
16 Carros Infantil R\$ 20,00
17 Galinha Pintadinha Infantil R\$ 20,00
18 No Limite Ação R\$ 30,00
19 Titanic Romance R\$ 18,00

De fato, os valores unitários foram copiados para a tabela principal. Aqueles que não constam na auxiliar mostraram um erro, o que era esperado. Porém perceba que o filme "Chamado" consta sim na tabela 2! Ocorreu o erro pois ao usarmos a alça de preenchimento, a região da tabela auxiliar também se deslocou para baixo. Devemos fixá-la usando o `$`:

`=PROCV(A6;B$15:D$19;3;0)`

Muitas vezes estaremos trabalhando com planilhas muito extensas e cheias de colunas. Na hora de aplicarmos a função `=PROCV()` podemos nos confundir na hora de passar a informação do número da coluna. Uma boa prática é numerá-las:

Tabela 2 - Lista de Produtos

Filme	Categoria	Valor
Chamado	Suspense	R\$ 42,00
Carros	Infantil	R\$ 20,00
Galinha Pintadinha	Infantil	R\$ 20,00
No Limite	Ação	R\$ 30,00
Titanic	Romance	R\$ 18,00

1 2 3

Lembrando sempre que a numeração começa pela primeira coluna da tabela e não de toda grade do documento do Excel.

Vejamos um outro exemplo para trabalharmos com a função `=PROCV()` :

E4	A	B	C	D	E	F	G	H	I	J	K	L	M
1	alura												
2	1	2	3	4	5								
3													
4	Matrícula	Unidade	Treinamento	Data	Aluno								
5	1010	Paulista	Word	10/01/2002	Alexandre								
6	3030	Itaim	Windows	18/01/2002	Karina								
7	2020	Vila Mariana	Excel	10/01/2002	Angelo								
8	4040	Vila Mariana	Power Point	10/01/2002	Camila								
9	8888	Paulista	Power Point	26/01/2002	Elisângela								
10	9090	Paulista	Windows	15/03/2002	Felipe								
11	1234	Paulista	Excel	15/03/2002	Karina								
12	4893	Itaim	Excel	23/05/2002	Luciana								
13													

Observamos uma planilha com alunos matriculados em alguns cursos, discriminando:

- Número de matrícula
- Unidade do treinamento
- O treinamento em si
- A data
- O nome do aluno

À direita temos uma ficha de consulta, onde podemos inserir esses dados. O que queremos aqui é poder inserir o número de matrícula em seu devido campo e o que o Excel traga o resto das informações para essa ficha. Perceba que a tabela da esquerda será a auxiliar, enquanto que a ficha será a principal. Como queremos procurar valores, é a `=PROCV()` que iremos utilizar.


Para facilitar a nossa vida, já foi colocada a numeração de cada coluna. Vamos também deixar delimitada a região da tabela auxiliar:

Região - A\$6:E\$20

Podemos dar um nome para essa região. Basta selecioná-la e, na *Caixa de Nome*, escrevermos o nome que queremos para ela:

	A	B	C	D	E
1	1	2	3	4	5
5	Matrícula	Unidade	Treinamento	Data	Aluno
6	1010	Paulista	Word	10/01/2002	Alexandre
7	3030	Itaim	Windows	18/01/2002	Karina
8	2020	Vila Mariana	Excel	10/01/2002	Ângelo
9	4040	Vila Mariana	Power Point	10/01/2002	Camila

Feito isso apertamos a tecla Enter. Se, por acaso, quisermos desfazer essa seleção, vamos na guia "FÓRMULAS" e em "Gerenciador de Nomes". Abrirá uma janela:



Aqui podemos excluir a seleção e começar de novo. Perceba que dando um nome para a região não precisamos nos preocupar mais com a fixação de linhas!

Comecemos, enfim, a implementar a função para trazer os dados para a ficha. Faremos primeiramente a parte da data:

```
=PROCV(célula com o número de matrícula na ficha; matriz_tabela; número da coluna de data; tipo de busca)
```

Logo, teremos:


```
=PROCV(H8;alunos;4,0)
```

	1	2	3	4	5
5	Matrícula	Unidade	Treinamento	Data	Aluno
6	1010	Paulista	Word	10/01/2002	Alexandre
7	3030	Itaim	Windows	18/01/2002	Karina
8	2020	Vila Mariana	Excel	10/01/2002	Ângelo
9	4040	Vila Mariana	Power Point	10/01/2002	Camila
10	8888	Paulista	Power Point	26/01/2002	Elisângela
11	9090	Paulista	Windows	15/03/2002	Felipe
12	1234	Paulista	Excel	15/03/2002	Karina
13	4893	Itaim	Excel	23/05/2002	Luciana
14	9092	Itaim	Word	27/07/2002	Patrícia
15	3432	Vila Mariana	Word	27/07/2002	Priscila

Consulta

Matrícula	3030	Inicio	37274
Nome Do Aluno	<input type="button" value=""/>		
Unidade	Treinamento		

O Excel inseriu a informação numérica pura, ou seja, o número que estamos visualizando atualmente na célula com a fórmula é a conta $18 / 1 / 2002$, o que traz um resultado esdrúxulo. Precisamos formatar a visualização do número para mostrar em forma de data:



Os formatos de data exibem números de série de data e hora como valores de data. Os formatos de data que começam com um asterisco (*) respondem a alterações nas configurações regionais de data e hora especificadas para o sistema operacional. Os formatos sem um asterisco não são afetados pelas configurações do sistema operacional.

Continuemos a aplicar a função para os outros campos:

- Nome do aluno: =PROCV(H8;alunos;5;0)
- Unidade: =PROCV(H8;alunos;2;0)
- Treinamento: =PROCV(H8;alunos;3;0)

De fato, facilmente a ficha é preenchida:

	1	2	3	4	5
4	Matrícula	Unidade	Treinamento	Data	Aluno
5					

Consulta

Podemos digitar qualquer matrícula válida, que o Excel nos retornará automaticamente os outros dados!

Busca por faixa numérica

Vejamos agora um exemplo para trabalharmos com busca por faixas numéricas, aquela que utilizaremos o `=PROCV()` tendo como o último argumento o `1` ou o `true` (verdadeiro):

5	Tabela 1		
6	Vendedor	Valor Vendido	Prêmio
7	Jose	R\$ 2.800,00	
8	Joao	R\$ 6.000,00	
9	Juca	R\$ 800,00	

Tabela 2	
Valor	Prêmio
	caneta
	relógio

Na primeira planilha vemos a listagem de vendedores de uma loja e quanto cada um vendeu em reais em um mês. O dono da loja quer presentear cada um de acordo com o valor vendido, como vemos na caixa azul. Na tabela 2 iremos preencher com as faixas numéricas dependendo do produto. Ela será nossa auxiliar, vamos prepará-la:

Perceba que cada prêmio está associado ao menor valor para ganhá-lo. O Excel entenderá que o valor máximo de cada faixa é o valor anterior ao da faixa seguinte. Daremos um nome para a tabela como fizemos no exemplo anterior para facilitar nossa vida: `premio`.

Comecemos a implementação da função:

```
=PROCV(célula com o valor vendido;matrix_tabela;número da coluna de prêmios;tipo de busca por faixa)
```

Portanto, para o primeiro vendedor, temos:

```
=PROCV(B7;premio;2;1)
```

Arrastamos usando a Alça de Preenchimento e completamos a tabela:

O mais importante na hora de implementar a função `=PROCV()` é lembrar de usar o *1* ou *true* como argumento e saber como montar a tabela auxiliar, utilizando as faixas numéricas corretamente.

Buscando informações em outras planilhas

Para o nosso último exemplo, temos esta planilha:

Estão discriminados diversos tipos de carros e precisamos preencher as outras colunas com a quantidade destes vendidos nos meses de Janeiro e Dezembro e o total de vendas. Mas perceba que não temos uma tabela auxiliar nesta aba, ela está em outra:

Esta mostra as quantidades vendidas no ano inteiro. Como buscar as informações de uma tabela que está em outra aba? Primeiramente vamos dar um nome a esta tabela auxiliar: `carros`. E já vamos guardar que a coluna de Janeiro é a *3* e a de Dezembro a *14*. Apliquemos a função na tabela principal:

```
=PROCV(célula com o número do item;matrix_tabela;número da coluna do mês;tipo de busca exata)
```

Então para Janeiro e Dezembro, respectivamente:

```
=PROCV(B3;carros;3;0)
```

```
=PROCV(B3;carros;14;0)
```

Como demos um nome para a tabela auxiliar, não importa se ela estiver em outra guia.

Para o total basta somar a células:

```
=D3+E3
```

Se quisermos, podemos aplicar diretamente a função `=PROCV()` na soma:

```
=PROCV(B3;carros;3;0)+PROCV(B3;carros;14;0)
```